

Exporting SAS/GRAPH® Output for Inclusion in Web Pages and Other Software Applications


Warren Repole Jr., SAS Institute Inc., Rockville MD

INTRODUCTION

When you execute SAS/GRAPH® code interactively, you typically view the picture on a display monitor through the GRAPH window. From the GRAPH window, you can export an image to a file that can be imported into other software applications or viewed through other interfaces such as a Web browser. However, when you have many images to export, this is not an efficient process. Instead, you can create the image files directly through a SAS/GRAPH program.

This paper covers the basic process for creating image files directly through a SAS/GRAPH program in SAS 9.1.3. Image types addressed in this paper include EMF, CGM, EPS, GIF, PNG, ActiveX, and PostScript. This paper also illustrates how those image files can be inserted into other software applications and Web pages.

SAS/GRAPH OUTPUT CREATION PROCESS



When SAS/GRAPH output is produced, each image is stored initially in a SAS catalog. The GRSEG entry type contains device-independent graphics commands. The default catalog name is `WORK.GSEG`.

Depending on the SAS/GRAPH device driver in effect through the `DEVICE=` option, device-specific commands are generated. When running SAS/GRAPH code interactively, the default device driver usually is a display monitor. For Windows, the default device driver is `WIN`. By default, interactive results are shown in the GRAPH window.

By selecting a device driver and establishing graphics options, image files can be created. An image file contains graphics commands that can be interpreted by an image renderer. The following common image renderers can process images created by SAS/GRAPH:

- Web browsers
- Microsoft Word and Microsoft PowerPoint
- Printers that support PostScript

A SAS/GRAPH image file is called a **graphics stream file**. The acronym **GSF** can be used to refer to a graphics stream file.

There are many reasons to create image files, including the following:

- You can import an image file into other software applications or display the image on a Web page.
- You might want to run SAS/GRAPH programs through batch or noninteractive processing.
- You can route output to a graphics device that is not attached to your SAS/GRAPH environment.
- You can move output to other operating environments.
- Your display device might have insufficient graphics display capabilities.

You can use the following methods to create image files:

- Export a single image from the GRAPH window through **File** → **Export as Image...**
- Create one or more graphics stream files using FILENAME and GOPTIONS statements.
- Create one or more image files using the Output Delivery System (ODS).

CHOOSING AN APPROPRIATE IMAGE FORMAT

You can create SAS/GRAPH output image files in many different graphics formats. For instance:

- GIF (Graphics Interchange Format)
- JPEG (Joint Photographic Experts Group)
- PNG (Portable Network Graphics)
- Windows Bitmap (BMP)
- Windows Metafile (WMF or EMF)
- CGM (Computer Graphics Metafile)
- PostScript (PS) and Encapsulated PostScript (EPS)

The ultimate destination of the image file determines an appropriate graphics format:

- Web pages typically display GIF, JPEG, and PNG images.
- Other image formats that are compatible with Microsoft Office applications include WMF, EMF, EPS, and CGM.

The desired image format usually determines an optimal SAS/GRAPH device driver.

Try to answer these questions when deciding upon the type of output file:

- Do you want to use system or hardware fonts that can be edited in the application?
- Do you plan to resize the graph in the target application?
- Should the graph have a transparent background?

Image Formats

There are two general classifications of image formats:

- Vector formats include EMF (Enhanced Windows Metafile), CGM, and EPS.
- Raster (or bitmap) formats include GIF, JPEG, PNG, and BMP.

The type of image format can determine your ability to resize and/or edit the image after the image is imported.

Vector formats can be resized without degrading the quality of the image. EMF and CGM images usually can be edited after being imported.

Raster formats can be resized but the quality of the image suffers. Raster images cannot be edited after being imported.

ActiveX is a Windows-based vector format.

- ActiveX SAS/GRAPH output can be generated as text within an HTML page.
- The text information is interpreted by a graph control (previously installed on the displaying machine) to render a graph dynamically.

One SAS/GRAPH implementation of ActiveX produces static images in the PNG raster format.

Detailed information on exporting SAS/GRAPH output for Microsoft Office applications is in the following SAS Technical Support document:

- TS-674, "An Introduction to Exporting SAS/Graph Output to Microsoft Office,"
<http://support.sas.com/techsup/technote/ts674/ts674.html>

USING PROC GDEVICE TO INVESTIGATE DEVICE DRIVERS

SAS/GRAPH Device Drivers

To determine what device drivers are available in your SAS/GRAPH installation, execute the GDEVICE procedure.

```
proc gdevice catalog=sashelp.devices;
```

SAS/GRAPH device drivers are stored in the **SASHELP.DEVICES** catalog.

The list of available device drivers is shown in the Directory window.

Name	Type	Description	Updated
ACTIVEX	DEV	ActiveX enabled GIF Driver	03/31/05
ACTXIMG	DEV	ActiveX enabled Image Driver	03/31/05
APLPLUS	DEV	Apple Laserwriter Plus (PostScript)	03/31/05
APPLELW	DEV	Apple Laserwriter (PostScript)	03/31/05
APPLELWM	DEV	Apple Laserwriter A4 paper	03/31/05
BMP	DEV	BMP File Format--256 colors	03/31/05
BMP20	DEV	BMP 2.0 File Format--256 colors	03/31/05
CAPSL3A4	DEV	Canon Laser Beam(CaPSL3) - A4 size	03/31/05
CAPSL3B5	DEV	Canon Laser Beam(CaPSL3) - B5 size	03/31/05
CAPSL3EX	DEV	Canon Laser Beam(CaPSL3) - Executive siz	03/31/05
CAPSL3LG	DEV	Canon Laser Beam(CaPSL3) - Legal size	03/31/05
CAPSL3LT	DEV	Canon Laser Beam(CaPSL3) - Letter size	03/31/05
CAPSL4A3	DEV	Canon Laser Beam(CaPSL4) - A3 size	03/31/05
CAPSL4A4	DEV	Canon Laser Beam(CaPSL4) - A4 size	03/31/05
CAPSL4A5	DEV	Canon Laser Beam(CaPSL4) - A5 size	03/31/05
CAPSL4B4	DEV	Canon Laser Beam(CaPSL4) - B4 size	03/31/05
CAPSL4B5	DEV	Canon Laser Beam(CaPSL4) - B5 size	03/31/05
CAPSL4EX	DEV	Canon Laser Beam(CaPSL4) - Executive siz	03/31/05
CAPSL4LG	DEV	Canon Laser Beam(CaPSL4) - Legal size	03/31/05
CAPSL4LS	DEV	Canon Laser Beam(CaPSL4) - Leisure size	03/31/05
CAPSL4LT	DEV	Canon Laser Beam(CaPSL4) - Letter size	03/31/05
CGM	DEV	CGM generator--binary output	03/31/05
CGMC	DEV	CGM generator w/colors--binary output	03/31/05
CGMCHAR	DEV	CGM generator--character format	03/31/05
CGMCLEAR	DEV	CGM generator--Clear text format	03/31/05
CGMLT97L	DEV	CGM for Smartsuite 97 landscape	03/31/05
CGMLT97P	DEV	CGM for Smartsuite 97 portrait	03/31/05
CGMOF97L	DEV	CGM for Office 97 landscape mode	03/31/05
CGMOF97P	DEV	CGM for Microsoft Office 97 portrait	03/31/05


Some default attributes of each device driver are shown in the Detail window.

GDEVICE: Detail	
Catalog: SASHELP.DEVICES	Entry: CGM
Orig Driver: CGM	Module: SASGDCGM
Description: CGM generator--binary output	Model: 3910
	Type: EXPORT
Lrows: 0	Xmax: 8.500 IN
Lcols: 0	Hsize: 0.000 IN
Prows: 100	Vsize: 0.000 IN
Pcols: 85	Horigin: 0.000 IN
	Vorigin: 0.000 IN
Aspect: 0.000	Rotate: _____
Driver query: _	Queued messages: N
	Paperfeed: 0.000 IN


EXPORTING IMAGES FROM THE GRAPH WINDOW

You can create an image file from a single graph displayed in the GRAPH window by completing these steps:

1. Select **File** → **Export as Image...**



2. Select the appropriate file type from the **Save as type** box.



3. Specify the desired filename.
4. Click **Save**.


The following image types can be created using the **File** → **Export as Image...** method:

- GIF (.gif)
- JPEG (.jpg)
- Bitmap (.bmp, .pbm, .dib)
- Windows Metafile (.emf, .wmf)
- Portable Network Graphics (.png)
- TIFF (.tif)
- PostScript (.ps)


INSERTING AN IMAGE FILE INTO AN APPLICATION OR WEB PAGE

Microsoft Office

To insert an image into a Microsoft Word document, select **Insert** → **Picture** → **from File...** from the menu bar.



Microsoft Word Document with Inserted GIF Image




HTML Documents

To manually insert an image into a Web page, add the IMG tag to the HTML document. The image filename is referenced in the SRC= attribute.

```
<IMG SRC="export.gif" >
```

Web Browser with Embedded GIF Image



CONCEPTS FOR GRAPHICS STREAM FILES

You can create image files by using SAS/GRAPH device drivers along with FILENAME and GOPTIONS statements.

- The FILENAME statement defines a fileref to be associated with one or more graphics stream files.
- The GOPTIONS statement specifies graphics options related to the creation of the graphics stream files.

SAS/GRAPH device drivers convert the device-independent catalog entry to standard graphics file formats such as GIF or EMF. Device drivers determine default values for the following attributes of the graphics output:

- dimensions and orientation
- text size
- colors
- hardware fonts used in the graph

To specify a device driver, use the DEVICE= option in the GOPTIONS statement:

```
goptions device=gif; /* GIF output */
goptions device=bmp; /* Windows Bitmap */
goptions device=wmf; /* Windows Metafile */
```

The following device drivers frequently are used to produce graphics stream files:

File Type	SAS/GRAPH Device Driver name
GIF files	GIF
JPEG files	JPEG
Windows Metafile	WMF
Enhanced Windows Metafile	EMF (Windows) SASEMF (UNIX)
Portable Network Graphics	PNG
PostScript (black and white, color)	PS, PSCOLOR
Computer Graphics Metafile	CGM

In addition to selecting a device driver, the following graphics options are used when creating graphics stream files:

GSFNAME=fileref

fileref associated with the graphics stream files

GSFMODE=REPLACE | APPEND

- specifies whether the records written should replace or be appended to the output file
- REPLACE is the default setting

CODING DETAILS FOR GRAPHICS STREAM FILES

Creating a Single File

Here are the preparation steps for creating a single GSF within a SAS/GRAPH program:

1. Use the FILENAME statement to define a fileref for the image file.
2. Assign the fileref to the GSFNAME= option in a GOPTIONS statement.
3. Use the default setting of GSFMODE=REPLACE.
4. Specify the device driver with the DEVICE= option.

To create a single graphics stream file, specify the filename in the FILENAME statement, for example:

```
filename grafout 'c:\temp\gsf\single.emf';
```

Here are the steps to complete the creation of the GSF:

1. Provide the SAS/GRAPH code to generate a single graph.
2. Use a FILENAME statement with the CLEAR option to release the fileref assigned to the GSF.

Program to Generate a Single Image File

This program produces a single GIF file:

```
filename grafout 'c:\temp\gsf\single.gif';
goptions reset=all gsfname=grafout gsfmode=replace device=gif;
proc gchart data=sashelp.class;
  vbar age / discrete;
  title 'Age Distribution for Students';
run;
quit;
filename grafout clear;
```

Creating Multiple Files

To create multiple image files, the FILENAME statement should point to an aggregate location such as a directory or folder, for example:

```
filename grafout 'c:\temp\gsf\';
```

Each image filename is based on the corresponding graphics catalog entry created by the SAS/GRAPH procedure.

Default names are given to graphics catalog entries created by SAS/GRAPH procedures:

- The first graph generated by a procedure is given the name of that procedure. For example, the first graph generated by PROC GPLOT is named GPLOT.
- Subsequent graphs generated by the same procedure are given the procedure name followed by an incrementing number. The second graph created by PROC GPLOT is GPLOT1, the third graph is named GPLOT2, and so on.

Program to Generate Multiple Image Files

This program produces two GIF files named **gchart.gif** and **gchart1.gif**:

```
filename grafout 'c:\temp\gsf\';
goptions reset=all gsfname=grafout gsfmode=replace device=gif;
title 'Student Demographics';
proc gchart data=sashelp.class;
  vbar age / discrete;
  hbar height / range;
run;
quit;
filename grafout clear;
```

CHOOSING NAMES FOR IMAGE FILES

You can use the NAME= option to name the graphs created by SAS/GRAPH procedures.

- For PROC GSLIDE and PROC GANNO, the NAME= option is used in the PROC statement.
- For procedures that use action statements to produce graphs, the NAME= option is used in the action statement such as PLOT, VBAR, or CHORO.

The NAME= value must be a valid SAS name of no more than 8 characters.

If you specify the NAME= option and a graph by that name already exists **in the output graphics catalog**, an incrementing number is added to the name.

- If you specify NAME="MYMAP" and a graph named MYMAP already exists in the catalog, the new graph is named MYMAP1.
- If you attempt to create yet another graph called MYMAP, the new graph is named MYMAP2 because MYMAP and MYMAP1 already exist.

Program to Generate Specific Image File Names

This program requests two GIF files named `ages.gif` and `heights.gif`:

```
filename grafout 'c:\temp\gsf\';
goptions reset=all gsfname=grafout gsfmode=replace device=gif;
title 'Student Demographics';
proc gchart data=sashelp.class;
  vbar age / discrete name="ages";
  hbar height / range name="heights";
run;
quit;
filename grafout clear;
```

Cleaning Out a Graphics Catalog

Be aware that SAS/GRAPH never replaces an existing graphics catalog entry. If you specify the NAME= option and a graph by that name already exists **in the output graphics catalog**, a unique name is generated based on your requested name. To delete existing graphs from a catalog, use the CATALOG, DATASETS, or GREPLAY procedure.

Each of these steps removes two graphs from the `WORK.GSEG` catalog:

```
proc greplay nofs igout=work.gseg;
  delete ages heights;
run;
quit;

proc catalog catalog=work.gseg;
  delete ages heights /
  entrytype=grseg;
run;
quit;
```

Each of these steps removes all graphs from the `WORK.GSEG` catalog:

```
proc greplay nofs igout=work.gseg;
  delete _all_;
run;
quit;

proc catalog catalog=work.gseg kill;
run;
quit;
```

This step deletes the entire `WORK.GSEG` catalog:

```
proc datasets lib=work nolist;
  delete gseg / memtype=catalog;
run;
quit;
```

CONTROLLING THE SIZE AND RESOLUTION OF IMAGES

Each device driver has maximum output dimensions and default output dimensions. Options declared in the GOPTIONS statement can change the output dimension settings.

The XMAX= option controls the maximum width of the graphics output area. The YMAX= option controls the maximum height of the graphics output area. The unit of measurement for these options can be inches, centimeters, or points. These options include any space required for a margin around the image.

The HSIZE= option controls the horizontal image size. The VSIZE= option controls the vertical image size. The unit of measurement for these options can be inches, centimeters, or points. These options are limited by the settings of the XMAX= and YMAX= options, respectively.

This program produces a single EMF image that is 6 inches wide and 4 inches high:

```
filename grafout 'c:\temp\gsf\inches.emf';
goptions reset=all gsfname=grafout gsfmode=replace device=emf
        hsize=6in vsize=4in;
proc gchart data=sashelp.class;
    vbar age / discrete;
    title 'Age Distribution for Students';
run;
quit;
filename grafout clear;
```

The XPIXELS= option controls the image width in pixels. The YPIXELS= option controls the image height in pixels. In conjunction with XMAX= and YMAX=, these options determine the resolution of the image.

This program produces a single GIF file that is 600 pixels wide and 400 pixels high:

```
filename grafout 'c:\temp\gsf\pixels.gif';
goptions reset=all gsfname=grafout gsfmode=replace device=gif
        xpixels=600 ypixels=400;
proc gchart data=sashelp.class;
    vbar age / discrete ;
    title 'Age Distribution for Students';
run;
quit;
filename grafout clear;
```

SPECIFYING TEXT FONTS IN IMAGE FILES

SAS/GRAPH software fonts are available in all operating environments. System fonts such as TrueType fonts often produce clearer text but might be available only in certain operating environments. Hardware fonts also typically produce clearer text and might apply to device drivers such as PostScript and SASEMF (the EMF driver for systems other than Windows).

This program produces a single EMF file that contains text written with a SAS/GRAPH font:

```
filename grafout 'c:\temp\gsf\sgfonts.emf';
goptions reset=all gsfname=grafout gsfmode=replace device=emf
        ftext=swiss;
proc gchart data=sashelp.class;
    vbar age / discrete;
    title 'Age Distribution for Students';
run;
quit;
filename grafout clear;
```

This program produces a single EMF file that contains text written with TrueType fonts:

```
filename grafout 'c:\temp\gsf\ttfonts.emf';
goptions reset=all gsfname=grafout gsfmode=replace device=emf
        ftext='Arial';
proc gchart data=sashelp.class;
    vbar age / discrete;
    title 'Age Distribution for Students';
run;
quit;
filename grafout clear;
```

This program produces a single EMF file that contains text written with hardware fonts:

```
filename grafout 'c:\temp\gsf\hwfonts.emf';
goptions reset=all gsfname=grafout gsfmode=replace device=sasemf
          ftext='Letter Gothic';
proc gchart data=sashelp.class;
  vbar age / discrete;
  title 'Age Distribution for Students';
run;
quit;
filename grafout clear;
```

USING ODS HTML TO CREATE IMAGE FILES FOR THE WEB

When the ODS HTML destination includes graphics output, images can be shown in one of two ways:


- Static images (GIF, JPEG, PNG) can be displayed within HTML pages using tags.
- Client graphics can be rendered through an ActiveX control or a Java applet.

Output files created with ODS differ from graphics stream files in several ways:

- ODS can interleave text output with graphics.
- ODS output can use ODS styles in conjunction with some device drivers.
- Graphics stream files contain only images. HTML files contain tags that point to static image files created by SAS/GRAPH.
- By default, the static images generated through ODS HTML are GIF files. Use the DEVICE= option to change the image file type.

Image files created through ODS can be used in conjunction with the HTML page. Those image files also can be used independently from the original HTML output. In general, specifying fonts and text heights is desirable when creating graphs for the Web.

The GPATH= option in the ODS HTML statement controls the storage location of the generated image files. If GPATH= is not specified, images are stored in the location indicated by the PATH= option. If the PATH= option also is omitted, images are stored in the default directory for your SAS session.



CODING THE PATH= AND GPATH= OPTIONS

If you intend to store the images in the same location as the generated HTML page, specify the PATH= option on the ODS HTML statement. If you intend to move the HTML file and its associated images to a new location after they are created, you **must** specify the URL=NONE suboption to synchronize the physical image location with the tags in the HTML page.

Program to Direct Image Files to the Same Location as the HTML File

This program creates an HTML page and a GIF image displayed within the page. All output files are stored in the same location:

```
ods listing close;
ods html path='c:\temp\gsf\' (url=none)
      body='odspath.html';
goptions reset=all ftext='Arial' htext=12pt;
proc gchart data=sashelp.class;
  hbar age / discrete name="odspath";
  title h=18pt 'Age Distribution';
run;
quit;
ods html close;
ods listing;
```

If you intend to store the images in a different location as the generated HTML page, specify the GPATH= option on the ODS HTML statement. To synchronize the physical image location with the tags in the HTML page, you **must** use the URL= suboption to specify the relative path between the HTML and the image files.

Program to Direct Image Files to a Different Location than the HTML File

This program stores an HTML page and a GIF image file in different locations:

```
ods listing close;
ods html path='c:\temp\gsf\'
      gpath='c:\temp\gsf\images\' (url='images/')
      body='odsgpath.html';
goptions reset=all ftext='Arial' htext=12pt;
proc gchart data=sashelp.class;
  hbar age / discrete name="odsgpath";
  title h=18pt 'Age Distribution';
run;
quit;
ods html close;
ods listing;
```

USING THE ACTXIMG AND JAVAIMG DRIVERS

Most device drivers used to produce static image files (such as GIF and EMF) do not support ODS styles in SAS 9.1.3. If you want to create a static image using ODS styles, use the ACTXIMG and JAVAIMG drivers. These drivers produce static PNG image files. Titles and footnotes are rendered as part of the HTML page. They do not appear in the static images.

The ACTXIMG device driver uses an ActiveX control to create PNG images. You must be running SAS/GRAPH in a Windows environment to use the ACTXIMG driver. The JAVAIMG device driver uses a Java applet to create PNG images. You can use the JAVAIMG driver in any operating environment that supports Java.


Program to Create Static PNG Images

This program stores an HTML page and a PNG image file in the same location:


```
ods html path='c:\temp\gsf\' (url=none)
      body='actximg.html'
      style=education;
goptions reset=all device=actximg;
proc gchart data=sashelp.class;
  vbar age / discrete name="actximg";
  title 'Age Distribution for Students';
run;
quit;
ods html close;
```

CREATING CLICKABLE GRAPHS FOR THE WEB

The ODS HTML destination supports the creation of clickable graphics.



Create a new variable that contains the HREF link to the associated Web page for each value of the chart variable.



This step creates the variable to be used for defining hyperlinks within a graph:

```
data prdsale_click;
  set sashelp.prdsale;
  length linkto $ 60;
  select(country);
    when('CANADA') linkto='HREF="http://www.canada.gc.ca/"';
    when('GERMANY') linkto='HREF="http://www.deutschland.de/"';
    when('U.S.A.') linkto='HREF="http://www.usa.gov/"';
    otherwise delete;
  end;
run;
```

When creating the graph, reference the new variable in the HTML= option in the action statement, for example:

```
pie country / html=linkto ;
```

This program creates a clickable pie chart:

```
ods listing close;
ods html path='c:\temp\gsf\' (url=None)
  body='clickpie.html';
options reset=all ftext='Arial' htext=12pt;
proc gchart data=prdsale_click;
  pie country / sumvar=Actual type=sum noheading
    html=linkto name="clickpie";
  format Actual dollar12.;
  title h=18pt 'Sales in Each Country';
run;
quit;
ods html close;
ods listing;
```

CREATING ACTIVEX AND JAVA GRAPHS

When you create graphs using drivers such as GIF and EMF, the graph is created as a static image.


- The image cannot be manipulated in real time by the user.
- Static graphs can have interactive capabilities such as data tips or drill-down links.

SAS/GRAPH can create graphs through ActiveX controls or Java applets. These graphs provide interactive features (via a pop-up menu) that allow you to modify the display as follows:


- zoom in and out
- subset the graph
- control properties such as colors, legends, and axes
- rotate the graph

ActiveX and Java graphics are referred to as *client graphics* because graphs are created through an ActiveX control or Java applet that must be installed on the client machine. After initial display, client graphs can be manipulated by the user. The ACTIVEX and JAVA drivers can be used with selected SAS/GRAPH procedures, including GCHART, GCONTOUR, GMAP, GPLOT, and G3D.

Static GIF Image File



ActiveX Graph (customized)



Creating a Single ActiveX Image


To create a graph through an ActiveX control, use the ODS HTML statement and specify ACTIVEX as the device driver:

```
ods html path='c:\temp\gsf\' (url=none)
         body='activex.html'
         style=education;
options reset=all device=activex;
proc gchart data=sashelp.class;
  vbar age / discrete;
title 'Age Distribution for Students';
run;
quit;
ods html close;
```


INSERTING AN ACTIVE X GRAPH INTO AN APPLICATION

Here are the primary steps to copy an ActiveX graph from a Web page into a Microsoft application:


1. Copy the ActiveX graph into the Windows clipboard by selecting **Copy** from the pop-up menu.



2. Insert the ActiveX graph image into the Microsoft application by selecting **Edit** → **Paste Special** from the menu bar.




3. Confirm the ActiveX graph insertion as a **SAS Graph v9 Control** by clicking **OK**.




Here are the primary steps to customize an ActiveX graph after insertion into a Microsoft application:


1. Activate the editing feature by selecting **SAS Graph v9 Object** → **Edit** from the pop-up menu.



2. Customize the ActiveX graph by using the selections available from the pop-up menu.



Customized ActiveX Graph



SPECIFICS FOR ENHANCED WINDOWS METAFILE (EMF) IMAGE FILES

EMF image files can be produced through multiple SAS/GRAPH device drivers, for example:.

- The EMF driver fully exploits system information only in Windows environments. [The EMF driver is available in other platforms, but acts like the SASEMF driver.]
- The SASEMF driver is available on all systems. [SASEMF is used to create graphics in ODS RTF output.]

EMF images are static files in vector format.

- EMF images can be resized after importing without degrading image quality.
- EMF images can be edited after importing.

The EMF driver recognizes TrueType fonts. The SASEMF driver supports three hardware fonts: Courier New, Courier, and Letter Gothic.

The EMF and SASEMF drivers have these default attributes.

Attribute	EMF Driver	SASEMF Driver
Resolution	96 pixels/inch	200 pixels/inch
Output Dimensions	600 pixels by 400 pixels (approximately 6 inches by 4 inches)	1200 pixels by 800 pixels (6 inches by 4 inches)

Special Considerations for EMF

In SAS 9.1.3, when using the FILENAME technique with an aggregate location, you must specify the NOFILEONLY option in the GOPTIONS statement. [See SAS Note 13115.]

SPECIFICS FOR COMPUTER GRAPHICS METAFILE (CGM) IMAGE FILES

CGM image files can be produced through numerous SAS/GRAPH device drivers for importing into Microsoft Office applications.

- The CGMOFML driver produces landscape output.
- The CGMOFMP driver produces portrait output.

CGM images are static files in vector format.

- CGM images can be resized after importing without degrading image quality.
- CGM images can be edited after importing.

The CGMOFML and CGMOFMP drivers support the Base-14 PostScript fonts as hardware fonts.

- The Base-14 PostScript fonts include Helvetica, Courier, TimesRoman, Symbol, and ZapfDingbats.
- In Microsoft Office, the TrueType font equivalents are Arial, Courier New, Times New Roman, Symbol, and Wingdings.

The CGM drivers have these default attributes:

- Resolution: 600 pixels/inch
- Output dimensions: Exactly 11 inches and 8.5 inches

Special Considerations for CGM

Multiple images can be stored within a single CGM file. However, only one image at a time can be imported into Microsoft Office.

SPECIFICS FOR ENCAPSULATED POSTSCRIPT (EPS) IMAGE FILES

EPS image files can be produced through multiple SAS/GRAPH device drivers.

- The PSLEPSF driver produces monochrome output.
- The PSLEPSC driver produces color output.

EPS images are static files in vector format.

- EPS images can be resized in other applications.
- EPS images **should not** be edited after importing.

The PSLEPSF and PSLEPSC drivers support all 35 standard PostScript fonts as hardware fonts.

- These PostScript fonts include Courier, Times, Helvetica, Helvetica Narrow, Palatino, New Century Schoolbook, ITC Bookman, IVC Avant Garde Gothic, ITC Zapf Chancery, Symbol, and ITC Zapf Dingbats.

The PSLEPSF and PSLEPSC drivers have these default attributes:

- Resolution: 300 pixels/inch
- Output dimensions: Exactly 11 inches and 8.5 inches
- Margins: Exactly 0.5 inch on all sides

SPECIFICS FOR GRAPHICS INTERCHANGE FORMAT (GIF) IMAGE FILES

GIF image files can be produced through multiple SAS/GRAPH device drivers.

- The GIF driver is the standard device driver. [GIF is used to create graphics in ODS HTML output.]
- Variations of the GIF driver create images of different sizes.

GIF images are static files in raster format.

- Resizing GIF images might decrease the image quality.
- GIF images cannot be edited after importing.

The GIF drivers recognize system fonts such as TrueType fonts on Windows. Because GIF output is rasterized, text might not appear as sharp as in vector format images.

The GIF driver has these default attributes:

- Resolution: 96 pixels/inch (cannot be altered)
- Output dimensions: 800 pixels by 600 pixels (approximately 8 inches by 6 inches)

Other GIF drivers produce images with widths of 733 pixels, 533 pixels, 373 pixels, 260 pixels, and 160 pixels. Those images also are produced with a 4:3 aspect ratio.


Special Considerations for GIF

The GIF driver supports transparency through the TRANSPARENCY graphic option. The value of the CBACK= option determines the transparent color. The GIFANIM driver creates animated GIF files composed of multiple images played in a "slide show" sequence.

This program produces a single transparent GIF image file:

```
filename grafout 'c:\temp\gsf\trans.gif';
goptions reset=all gsfname=grafout gsfmode=replace
          device=gif transparency;
proc gchart data=sashelp.class;
  vbar age / discrete;
  title 'Age Distribution for Students';
run;
quit;
filename grafout clear;
```

Difference between a Regular GIF Image and a Transparent GIF Image



This program produces an animated GIF image file:

```
filename grafout 'c:\temp\gsf\gifanim.gif';
goptions reset=all gsfname=grafout gsfmode=replace device=gifanim
          delay=300 /* hundredths of a second between images */
          iteration=5 /* times through the loop (0=endless)*/ ;
title 'Student Demographics';
proc gchart data=sashelp.class;
  vbar age / discrete;
run;
goptions gsfmode=append;
  hbar height / range;
run;
goptions gepilog='3B'x;
  pie weight / type=percent;
run;
quit;
filename grafout clear;
```

SPECIFICS FOR PORTABLE NETWORK GRAPHICS (PNG) IMAGE FILES

PNG image files can be produced through multiple SAS/GRAPH device drivers.

- The PNG driver is available on all systems.
- The ACTXIMG driver is available only under Windows and requires the use of the ODS HTML destination.

PNG images are static files in raster format.

- Resizing PNG images might decrease the image quality.
- PNG images cannot be edited after importing.

The PNG and ACTXIMG drivers recognize TrueType fonts. Because PNG output is rasterized, text might not appear as sharp as in vector format images.

The PNG and ACTXIMG drivers have these default attributes.

Attribute	PNG Driver	ACTXIMG Driver
Resolution	96 pixels/inch	96 pixels/inch
Output Dimensions	600 pixels by 350 pixels (approximately 6 inches by 3.5 inches)	640 pixels by 480 pixels (approximately 6.5 inches by 5 inches)
Transparency	Not supported [See SAS Note 12442]	Not supported
Location of Titles	Within the image	Outside the image (in the HTML document)
ODS Styles	Cannot be applied	Can be applied

Special Considerations for PNG

In SAS 9.1.3, when using the FILENAME technique with an aggregate location, you must specify the NOFILEONLY option in the GOPTIONS statement. [See SAS Note 13115.]

SPECIFICS FOR WINDOWS BITMAP (BMP) IMAGE FILES

BMP image files can be produced through the BMP SAS/GRAPH device driver. BMP images are static files in raster format.

- Resizing BMP images can decrease the image quality.
- BMP images cannot be edited after importing.

BMP images are not compressed, so the size of the output file can be very large.

The BMP device driver recognizes TrueType fonts. Because BMP output is rasterized, text might not appear as sharp as in vector format images.

The BMP driver has these default attributes:

- Resolution: 96 pixels/inch
- 600 pixels by 400 pixels (approximately 6 inches by 4 inches)


SPECIFICS FOR POSTSCRIPT (PS) IMAGE FILES

PostScript image files can be produced through multiple SAS/GRAPH device drivers.

- The PS driver produces monochrome output.
- The PSCOLOR driver produces color output.

PostScript images are static files in vector format.

Multiple images can be stored in one PS file. Each image appears on a separate page of printed PostScript output.



The PS and PSCOLOR drivers support all 35 standard PostScript fonts as hardware fonts.

- These PostScript fonts include Courier, Times, Helvetica, Helvetica Narrow, Palatino, New Century Schoolbook, ITC Bookman, IVC Avant Garde Gothic, ITC Zapf Chancery, Symbol, and ITC Zapf Dingbats.

The PS and PSCOLOR drivers have these default attributes:

- Resolution: 300 pixels/inch
- Output dimensions: Exactly 11 inches and 8.5 inches
- Margins: Exactly 0.5 inch on all sides

SPECIFICS FOR THE ACTIVEX AND JAVA DEVICE DRIVERS

The ACTIVEX device driver uses an ActiveX control to create graphs. You must be running SAS/GRAPH in a Windows environment to use the ACTXIMG driver. The JAVA device driver uses a Java applet to create graphs. You can use the JAVA driver in any operating environment that supports Java.

The ActiveX control must be installed on the client machine in order to display a graph created with the ACTIVEX driver. You can manually download and install the ActiveX control from the SAS web site. [Go to <http://support.sas.com/rnd/samples/graph/activex.html>] You can specify a URL for the ActiveX control through the CODEBASE= option in the ODS HTML statement.

- If the ActiveX control was not previously installed, the user will be prompted to install it.

The Java plug-in from Sun Microsystems must be installed on the client machine in order to display a graph created with the JAVA driver.

- On Windows systems, the user is prompted to install the plug-in if it is not already installed.
- On other systems, the plug-in can be installed from the Sun Microsystems Web site. [Go to <http://www.sun.com>]

The CODEBASE= option in the ODS HTML statement points to the URL of the required SAS/GRAPH Java archive file. [This is not the same as the CODEBASE= parameter of the ATTRIBUTES= option.]

Graphics created using client drivers use ODS styles to determine default colors and fonts. Thus, graphs might look different than graphs created with other drivers using the same code. Because the graph is created external to SAS/GRAPH, some procedure-specific options are not supported. Many options in GOPTIONS, AXIS, LEGEND, and SYMBOL statements are not supported.

Titles and footnotes are generated as part of the Web page and not part of the graph itself. Default height, font, and color are based on the ODS style. Changes to the default height depend on the height unit associated with the text.

If you specify a text height without specifying a unit, the height is interpreted as a **multiple** of the default text height for the ODS style. For example, if the default height of title text is 16 points, the following TITLE statement generates text with a height of 160 points.

```
title h=10 'This title is 160 points high';
```

If you specify a text height with a unit of PERCENT (PCT), the height is interpreted as a **percentage** of the default text height for the ODS style. For example, if the default height of title text is 16 points, the following TITLE statement generates text with a height of 1.6 points.

```
title h=10pct 'This title is 1.6 points high';
```

ActiveX images are vector graphs that are generated by a viewing application such as Internet Explorer, Microsoft Word, and Microsoft PowerPoint.

- ActiveX images can be resized after importing.
- Editing is accomplished through user interaction with the graph in the context of the viewing application.

Output dimensions of ActiveX output default to 640 pixels by 480 pixels.

CONCLUSION

The concepts and techniques covered in this paper can assist you in creating image files directly through a SAS/GRAPH program. Saving graphs as image files allows you to insert those images into other software applications and Web pages.

REFERENCES

SAS Institute Inc. 2005. *An Introduction to Exporting SAS/Graph Output to Microsoft Office*.

<<http://support.sas.com/techsup/technote/ts674/ts674.html>>

SAS Institute Inc. 2007. *SAS® Color Graphics Course Notes*. Cary, NC: SAS Institute Inc.

SAS Institute Inc. 2004. *SAS/GRAPH® 9.1 Reference, Volumes 1 and 2*. Cary, NC: SAS Institute Inc.

<http://support.sas.com/documentation/onlinedoc/91pdf/sasdoc_91/graph_ref_6960.pdf>

Schrader, Dawn. 2003. "Exporting SAS/GRAPH Output: Concepts and Ideas." *Proceedings of the Twenty-Eighth Annual SAS® Users Group International Conference*. Cary, NC: SAS Institute Inc.

<<http://www2.sas.com/proceedings/sugi28/139-28.pdf>>

CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the author at:

Warren Repole Jr.
SAS Training Center
111 Rockville Pike Ste 1000
Rockville MD 20850
warren.repole@sas.com

TRADEMARK NOTICE

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.

Other brand and product names are registered trademarks or trademarks of their respective companies.